Quantum Runge-Lenz Vector and the Hydrogen Atom, the hidden SO(4) symmetry using Computer Algebra

Pascal Szriftgiser[1] and Edgardo S. Cheb-Terrab[2]
(1) University of Lille, CNRS, UMR 8523 - PhLAM - Physique des Lasers, Atomes et Molécules, F-59000 Lille, France
(2) Maplesoft

Abstract

Pauli first noticed the hidden SO(4) symmetry for the Hydrogen atom in the early stages of quantum mechanics [1]. Departing from that symmetry, one can recover the spectrum of a spinless hydrogen atom and the degeneracy of its states without explicitly solving Schrödinger's equation [2]. In this paper, we derive that SO(4) symmetry and spectrum using a computer algebra system (CAS). While this problem is well known [3, 4], its solution involves several steps of manipulating expressions with tensorial quantum operators, simplifying them by taking into account a combination of commutator rules and Einstein's sum rule for repeated indices. Therefore, it is an excellent model to test the current status of CAS concerning this kind of quantum-and-tensor-algebra computations. Generally speaking, when capable, CAS can significantly help with manipulations that, like non-commutative tensor calculus subject to algebra rules, are tedious, time-consuming and error-prone. The presentation also shows a pattern of computer algebra operations that can be useful for systematically tackling more complicated symbolic problems of this kind.

Introduction

The primary purpose of this work is to derive, step-by-step, the SO(4) symmetry of the Hydrogen atom and its spectrum using a computer algebra system (CAS). To the best of our knowledge, such a derivation using symbolic computation has not been shown before. Part of the goal was also to see whether this computation can be performed entering only the main definition formulas, followed by only simplification commands, and without using previous knowledge of the result. The intricacy of this problem is in the symbolic manipulation and simplification of expressions involving noncommutative quantum tensor operators. The simplifications need to take into account commutator rules, symmetries under permutation of indices of tensorial subexpressions, and use Einstein's sum rule for repeated indices.

We performed the derivation using the Maple 2020 system with the Maplesoft Physics Updates v.705. Generally speaking, the default computational domain of CAS doesn't include tensors, noncommutative operators nor related simplifications. On the other hand, the Maple system is distributed with a Physics package that extends that default domain to include those objects and related operations. Physics includes a Simplify command that takes into account custom algebra rules and the sum rule for repeated indices, and uses tensor-simplification algorithms [5] extended to the noncommutative domain.

A note about notation: when working with a CAS, besides the expectation of achieving a correct result for a complicated symbolic calculation, readability is also an issue. It is desired that one be able to enter the definition formulas and computational steps to be performed (the *input*, preceded by a prompt >, displayed in black) in a way that resembles as closely as possible their paper and pencil representation, and that the results (the *output*, computed by Maple, displayed in blue) use textbook mathematical-physics notation. The Physics package implements such dedicated typesetting. In what follows, within text and in the *output*, noncommutative objects are displayed using a different color, e.g. H, vectors and tensor indices are displayed the standard way, as in $\vec{L}$, and L_q, and commutators are displayed with a minus subscript, e.g. $[H, L_q]_-$. Although the Maple system allows for providing dedicated typesetting also for the *input*, we preferred to keep visible the Maple *input* syntax, allowing for comparison with paper and pencil notation. We collected the names of the commands used and a one line description for them in an Appendix at the end. Maple also implements the concept of *inert* representations of computations, which are activated only when desired. We use this feature in several places. Inert computations are entered by preceding the command with % and are displayed in grey. Finally, as is usual in CAS, every output has an equation label, which we use throughout the presentation to refer to previous intermediate results.

In Sec.1, we recall the standard formulation of the problem and present the computational goal, which is the derivation of the formulas representing the SO(4) symmetry and related spectrum.

In Sec.2, we set tensorial non-commutative operators representing position and linear and angular momentum, respectively X_a, p_a and L_a, their commutation rules used as departure point, and the form of the quantum Hamiltonian H. We also derive a few related identities used in the sections that follow.

In Sec.3, we derive the conservation of both angular momentum and the Runge-Lenz quantum operator, respectively

$\left[H, L_q\right]_- = 0$ and $\left[H, Z_k\right]_- = 0$. Taking advantage of the *differentialoperators* functionality in the Physics package, we perform the derivation exploring two equivalent approaches; first using only a symbolic tensor representation p_j of the momentum operator, then using an explicit differential operator representation for it in configuration space, $p_j = -i\,\hbar\,\partial_j$. With the first approach, expressions are simplified only using the departing commutation rules and Einstein's sum rule for repeated indices. Using the second approach, the problem is additionally transformed into one where the differentiation operators are applied explicitly to a test function $G(X)$. Presenting both approaches is of potential interest as it offers two partly independent methods for performing the same computation, which is helpful to provide confidence on in the results when unknown, a relevant issue when using computer algebra.

In Sec. 4, we derive $\left[L_m, Z_n\right]_- = \hbar\,\mathrm{i}\,\epsilon_{m,n,u}\,Z_u$ and show that the classical relation between angular momentum and the Runge-Lenz vectors, $\vec{L}\cdot\vec{Z} = 0$, due to the orbital momentum being perpendicular to the elliptic plane of motion while the Runge-Lenz vector lies in that plane, still holds in quantum mechanics, where the components of these quantum vector operators do not commute but $\vec{L}\cdot\vec{Z} = \vec{Z}\cdot\vec{L} = 0$.

In Sec. 5, we derive $\left[Z_a, Z_b\right]_- = -\dfrac{2\,\mathrm{i}\,\hbar\,\epsilon_{a,b,c}\left(H\,L_c\right)}{m_e}$ using the two alternative approaches described for Sec.3.

In Sec. 6, we derive the well-known formula for the square of the Runge-Lenz vector, $Z_k^{\,2} = \dfrac{2\,H\left(\hbar^2 + L_a^{\,2}\right)}{m_e} + \kappa^2$.

Finally, in Sec. 7, we use the SO(4) algebra derived in the previous sections to obtain the spectrum of the Hydrogen atom. Following the literature, this approach is limited to the bound states for which the energy is negative.

Some concluding remarks are presented at the end, and input syntax details are summarized in an Appendix.

A Maple worksheet containing this presentation can be downloaded from https://www.mapleprimes.com/posts/208810-The-Hidden-SO4-Symmetry-Of-The-Hydrogen-Atom.

1. The hidden SO(4) symmetry of the Hydrogen atom

Let's consider the Hydrogen atom and its Hamiltonian

$$H = \frac{\|\vec{p}\|^2}{2\,m_e} - \frac{\kappa}{r},$$

where $\vec{p}$ is the electron momentum, m_e its mass, κ a real positive constant, $r = \|\vec{r}\| \equiv \sqrt{X_a^{\,2}}$ the distance of the electron from the proton located at the origin, and X_a is its tensorial representation with components $[x, y, z]$. We assume that the proton's mass is infinite. The electron and nucleus spin are not taken into account. Classically, from the potential $-\dfrac{\kappa}{r}$, one can derive a central force $\vec{F} = -\kappa\,\dfrac{\hat{r}}{r^2}$ that drives the electron's motion. Introducing the angular momentum

$$\vec{L} = \vec{r}\times\vec{p},$$

one can further define the Runge-Lenz vector $\vec{Z}$:

$$\vec{Z} = \frac{1}{m_e}\,\vec{L}\times\vec{p} + \kappa\,\frac{\vec{r}}{r}.$$

It is well known that $\vec{Z}$ is a constant of the motion, i.e. $\dfrac{\mathrm{d}}{\mathrm{d}t}\,\vec{Z}(t) = 0$. Switching to Quantum Mechanics, this condition reads

$$\left[H, \vec{Z}\right]_- = 0.$$

where, for hermiticity purpose, the expression of $\vec{Z}$ must be symmetrized

$$\vec{Z} = \frac{1}{2\,m_e}\left(\vec{L}\times\vec{p} - \vec{p}\times\vec{L}\right) + \kappa\,\frac{\vec{r}}{r}.$$

In what follows, departing from the Hamiltonian H, the basic commutation rules between position $\vec{r}$, momentum $\vec{p}$ and angular momentum $\vec{L}$ in tensor notation, we derive the following commutation rules between the quantum Hamiltonian, angular momentum and Runge-Lenz vector $\vec{Z}$

$$\begin{aligned}
\left[H, L_n\right]_- &= 0\\
\left[H, Z_n\right]_- &= 0\\
\left[L_m, Z_n\right]_- &= \mathrm{i}\,\hbar\,\epsilon_{m,n,o}\,Z_o\\
\left[Z_m, Z_n\right]_- &= -2\,\frac{\mathrm{i}\,\hbar}{m_e}\,H\,\epsilon_{m,n,o}\,L_o
\end{aligned}$$

Since H commutes with both $\vec{L}$ and $\vec{Z}$, defining

$$M_n = \sqrt{-\frac{m_e}{2\,H}}\;Z_n,$$

these commutation rules can be rewritten as

$$\begin{aligned}
\left[L_m, L_n\right]_- &= \mathrm{i}\,\hbar\,\epsilon_{m,n,o}\,L_o\\
\left[L_m, M_n\right]_- &= \mathrm{i}\,\hbar\,\epsilon_{m,n,o}\,M_o\\
\left[M_m, M_n\right]_- &= \mathrm{i}\,\hbar\,\epsilon_{m,n,o}\,L_o
\end{aligned}$$

This set constitutes the Lie algebra of the SO(4) group.

2. Setting the problem, commutation rules and useful identities

Load the *Physics* package and its *Library* subpackage containing additional manipulation commands; set the imaginary unit to be represented by a lowercase Latin i letter.

```
> restart;
  with(Physics) : with(Library) :
  interface(imaginaryunit = i) :
```

Set the context: Cartesian coordinates, 3D Euclidean space, lowercase letters representing tensor indices, use automatic simplification (automatically *simplify the size* of everything being displayed) and indicate that all of $\{\hbar, \kappa, m_e\}$ are real objects.

```
> Setup(coordinates = cartesian, realobjects = {ℏ, κ, m_e}, automaticsimplification = true, dimension = 3, metric
      = Euclidean, spacetimeindices = lowercaselatin, quiet)
```

$$\left[automaticsimplification = true,\ coordinatesystems = \{X\},\ dimension = 3,\ metric = \{(1, 1) = 1, (2, 2) = 1, (3, 3) = 1\},\ realobjects = \{\hbar, \kappa, m_e, x, y, z\},\ spacetimeindices = lowercaselatin\right] \quad \textbf{(1)}$$

Set quantum Hermitian operators (not Z, we derive that property for it further below) and related commutators:

- The dimensionless potential $V = \frac{1}{r}$ is assumed to commute with position, not with momentum - the commutation rule with p is derived in Sec.2.2.
- The commutator rules between position X_n on the one hand, and linear p_k and angular momentum L_k are the departure point, entered using the inert form of the *Commutator* command. Tensors are indexed using the standard Maple indexation []
.

```
> Setup(quantumoperators = {Z},
   hermitianoperators = {V, H, L, X, p},
   algebrarules = {
      %Commutator(p[k], p[l]) = 0,
      %Commutator(X[k], p[l]) = i·ℏ·g_[k, l],
      %Commutator(L[j], L[k]) = i·ℏ·LeviCivita[j, k, n]·L[n],
      %Commutator(p[j], L[k]) = i·ℏ·LeviCivita[j, k, n]·p[n],
      %Commutator(X[j], L[k]) = i·ℏ·LeviCivita[j, k, n]·X[n],
      %Commutator(X[k], V(X)) = 0})
```

$$\left[algebrarules = \left\{ [L_j, L_k]_- = i\,\hbar\,\epsilon_{j,k,n} L_n,\ [p_j, L_k]_- = i\,\hbar\,\epsilon_{j,k,n} p_n,\ [p_k, p_l]_- = 0,\ [X_j, L_k]_- = i\,\hbar\,\epsilon_{j,k,n} X_n,\ [X_k, p_l]_- = i\,\hbar\, g_{k,l},\ [X_k, V(X)]_- = 0 \right\},\ hermitianoperators = \{H, L, V, p, x, y, z\},\ quantumoperators = \{H, L, V, Z, p, x, y, z\}\right] \quad \textbf{(2)}$$

Define the *tensor* quantum operators representing the linear momentum, angular momentum and the Runge-Lenz vectors

```
> Define(p[k] = [p_x, p_y, p_z], L[k] = [L_x, L_y, L_z], Z[k] = [Z_x, Z_y, Z_z], quiet)
```

$$\left\{\gamma_a, L_k, \sigma_a, Z_k, \partial_a, g_{a,b}, p_k, \epsilon_{a,b,c}, X_a\right\} \quad \textbf{(3)}$$

For readability, avoid redundant display of functionality

```
> CompactDisplay(V(X))
```

$$V(X) \textit{ will now be displayed as } V \quad \textbf{(4)}$$

The Hamiltonian for the hydrogen atom is entered as

```
> H = p[l]^2/(2·m_e) − κ·V(X)
```

$$H = \frac{p_l^2}{2\, m_e} - \kappa\, V \quad \textbf{(5)}$$

2.1 Definition of $V(X)$ and related identities

We use the dimensionless potential $V(X)$

```
> V(X) = 1/(X[l]^2)^(1/2)
```

$$V = \left(X_l^2\right)^{-\frac{1}{2}} \quad \textbf{(6)}$$

The gradient of $V(X)$ is

```
> d_[n]((6))
```

$$\partial_n(V) = -\left(X_l^2\right)^{-\frac{3}{2}} X_n \quad \textbf{(7)}$$

where we note that all these commands (including product and power), *distribute over equations*. So that

```
> subs((rhs = lhs)((6)^3), (7))
```

$$\partial_n(V) = -V^3 X_n \quad \textbf{(8)}$$

Equivalently, from **(6)** one can deduce $V^3 X_l^2 = V$ that will often be used afterwards

> $(rhs = lhs)\left(\frac{V(X)^3}{\mathbf{(6)}^2}\right)$

$$V^3\, X_l^2 = V \tag{9}$$

2.2 The commutation rules between $\vec{L}$, $\vec{p}$ and the potential $V(X)$

By definition,

> $L[q] = LeviCivita[q, m, n] \cdot X[m] \cdot p[n]$

$$L_q = \epsilon_{m,n,q}\, X_m\, p_n \tag{10}$$

so

> $Commutator(\mathbf{(10)}, V(X))$

$$[L_q, V]_- = \epsilon_{m,n,q}\, X_m\, [p_n, V]_- \tag{11}$$

The commutator on the right-hand side cannot be computed until providing more information. To derive the value of $[p_n, V]_-$ we introduce an arbitrary test function $G(X)$, and set p_n as a *differentialoperator*

> $Setup(differentialoperators = \{[p[k], [x, y, z]]\})$

$$[differentialoperators = \{[p_k, [X]]\}] \tag{12}$$

Now, apply to $G(X)$ the differential operator p_n found in the commutator of the right-hand side of **(11)**

> $(lhs = ApplyProductsOfDifferentialOperators@rhs)(\mathbf{(11)} \cdot G(X))$

$$[L_q, V]_-\, G(X) = \epsilon_{m,n,q}\, X_m \left(p_n(V\, G(X)) - V\, p_n(G(X))\right) \tag{13}$$

The result of $p_l(G(X))$ is not known to the system at this point. Define then an explicit representation for p_n as the differential operator in configuration space $p_n = -i\, \hbar\, \partial_n$

> $p := u \rightarrow -i\, \hbar \cdot d\_[op(procname)](u)$

$$p := u \mapsto -\mathrm{i}\, \hbar\, \partial_{op(procname)}(u) \tag{14}$$

With this definition, we can compute the commutator in **(11)**

> **(13)**

$$[L_q, V]_-\, G(X) = -\mathrm{i}\, \epsilon_{m,n,q}\, \hbar\, X_m\, \partial_n(V)\, G(X) \tag{15}$$

So that using **(8)** $\equiv \partial_n(V) = -V^3\, X_n$ and multiplying by $G(X)^{-1}$,

> $SubstituteTensor(\mathbf{(8)}, \mathbf{(15)}) \cdot G(X)^{-1}$

$$[L_q, V]_- = \mathrm{i}\, \epsilon_{m,n,q}\, \hbar\, X_m\, V^3\, X_n \tag{16}$$

from where we get the first commutation rule:

> $Simplify(\mathbf{(16)})$

$$[L_q, V]_- = 0 \tag{17}$$

Likewise, from the *inert* = *active* form of $[p_q, V(X)]_-$

> $(\%Commutator = Commutator)(p[q], V(X))$

$$[p_q, V]_- = [p_q, V]_- \tag{18}$$

by applying this equation to the test function $G(X)$ we get

> $(lhs = ApplyProductsOfDifferentialOperators@rhs)(\mathbf{(18)} \cdot G(X))$

$$[p_q, V]_-\, G(X) = -\mathrm{i}\, \hbar\, \partial_q(V)\, G(X) \tag{19}$$

> $SubstituteTensor(\mathbf{(8)}, \mathbf{(19)}) \cdot G(X)^{-1}$

$$[p_q, V]_- = \mathrm{i}\, \hbar\, V^3\, X_q \tag{20}$$

In the same way, for $[p_q, V^3]_-$ we get

> $(\%Commutator = Commutator)(p[q], V(X)^3)$

$$\left[p_q, V^3\right]_- = \left[p_q, V^3\right]_- \quad (21)$$

> $(lhs = ApplyProductsOfDifferentialOperators@rhs)((\mathbf{21}) \cdot G(X))$

$$\left[p_q, V^3\right]_- G(X) = -\mathrm{i}\,\hbar \left(\partial_q(V)\, V^2 + V\, \partial_q(V)\, V + V^2\, \partial_q(V)\right) G(X) \quad (22)$$

> $SubstituteTensor((\mathbf{8}), (\mathbf{22})) \cdot G(X)^{-1}$

$$\left[p_q, V^3\right]_- = \mathrm{i}\,\hbar \left(V^3 X_q V^2 + V^4 X_q V + V^5 X_q\right) \quad (23)$$

> $(lhs = Simplify@rhs)((\mathbf{23}))$

$$\left[p_q, V^3\right]_- = 3\,\mathrm{i}\,\hbar\, V^5 X_q \quad (24)$$

Add now these new commutation rules to the setup of the problem so that they are taken into account when using *Simplify*

> **(17)**, **(20)**, **(24)**

$$\left[L_q, V\right]_- = 0,\ \left[p_q, V\right]_- = \mathrm{i}\,\hbar\, V^3 X_q,\ \left[p_q, V^3\right]_- = 3\,\mathrm{i}\,\hbar\, V^5 X_q \quad (25)$$

> $Setup((\mathbf{25}))$

$$\left[algebrarules = \left\{\left[L_j, L_k\right]_- = \mathrm{i}\,\hbar\,\epsilon_{j,k,n} L_n,\ \left[L_q, V\right]_- = 0,\ \left[p_j, L_k\right]_- = \mathrm{i}\,\hbar\,\epsilon_{j,k,n} p_n,\ \left[p_k, p_l\right]_- = 0,\ \left[p_q, V\right]_- = \mathrm{i}\,\hbar\, V^3 X_q,\ \left[p_q, V^3\right]_- = 3\,\mathrm{i}\,\hbar\, V^5 X_q,\ \left[X_j, L_k\right]_- = \mathrm{i}\,\hbar\,\epsilon_{j,k,n} X_n,\ \left[X_k, p_l\right]_- = \mathrm{i}\,\hbar\, g_{k,l},\ \left[X_k, V\right]_- = 0\right\}\right] \quad (26)$$

Undo *differentialoperators* to work using two different approaches, with and without them.

> $Setup(differentialoperators = none)$

$$[differentialoperators = none] \quad (27)$$

3. Commutation rules between the Hamiltonian and each of the angular momentum and Runge-Lenz tensors

Departing from the Hamiltonian of the hydrogen atom **(5)** and the definition of angular momentum **(10)**

> **(5)**; **(10)**;

$$H = \frac{p_l^2}{2\, m_e} - \kappa\, V$$

$$L_q = \epsilon_{m,n,q} X_m p_n \quad (28)$$

by taking their commutator we get

> $Commutator((\mathbf{5}), (\mathbf{10}))$

$$\left[H, L_q\right]_- = \frac{-\mathrm{i}\,\epsilon_{m,n,q}\,\hbar \left(-X_m V^3 X_n \kappa\, m_e + p_l p_n g_{l,m}\right)}{m_e} \quad (29)$$

> $Simplify((\mathbf{29}))$

$$\left[H, L_q\right]_- = 0 \quad (30)$$

3.1 $\left[H, Z_n\right]_- = 0$, algebraic approach

Start from the definition of the quantum Runge-Lenz tensor

> $Z[k] = \frac{1}{2\, m_e} \cdot LeviCivita[a, b, k] \cdot (L[a] \cdot p[b] - p[a] \cdot L[b]) + \kappa \cdot V(X) \cdot X[k]$

$$Z_k = \frac{\epsilon_{a,b,k}\left(L_a p_b - p_a L_b\right)}{2\, m_e} + \kappa\, V X_k \quad (31)$$

This tensor is Hermitian

> **(31)** $-$ $Dagger((\mathbf{31}))$

$$Z_k - Z_k^{\dagger} = \frac{2\,\kappa\, V X_k\, m_e - 2\,\kappa\, X_k\, V\, m_e + \epsilon_{a,b,k}\left(L_a\, p_b + L_b\, p_a - p_a\, L_b - p_b\, L_a\right)}{2\, m_e} \tag{32}$$

> *Simplify*(**(32)**)

$$Z_k - Z_k^{\dagger} = 0 \tag{33}$$

Since the system knows about the commutation rule between linear and angular momentum,

> (*%Commutator* = *Commutator*)(*L*[*a*], *p*[*b*])

$$\left[L_a, p_b\right]_- = \mathrm{i}\,\hbar\,\epsilon_{a,b,n}\, p_n \tag{34}$$

the expression **(31)** for Z_k can be simplified

> *Simplify*(**(31)**)

$$Z_k = \frac{\mathrm{i}\,\hbar\, p_k}{m_e} + \kappa\, V X_k - \frac{\epsilon_{a,b,k}\, p_a\, L_b}{m_e} \tag{35}$$

and the angular momentum removed from the the right-hand side using **(10)** ≡ $L_q = \epsilon_{m,n,q}\, X_m\, p_n$, so that Z_k gets expressed entirely in terms of p_k, X and V

> *Simplify*(*SubstituteTensor*(**(10)**, **(35)**))

$$Z_k = \frac{-\mathrm{i}\,\hbar\, p_k + \kappa\, V X_k\, m_e + X_m\, p_k\, p_m - X_k\, p_m^2}{m_e} \tag{36}$$

Taking the commutator between **(5)** ≡ $H = \frac{p_l^2}{2\, m_e} - \kappa\, V$, and this expression for Z_k we have the starting point towards showing that $\left[H, Z_k\right]_- = 0$

> *Simplify*(*Commutator*(**(5)**, **(36)**))

$$\left[H, Z_k\right]_- = \frac{\kappa\,\hbar\left(\hbar\, V^5 X_a^2\, X_k + \hbar\, V^3 X_k - 2\,\mathrm{i}\, X_a\, X_k\, p_a\, V^3 - 2\,\mathrm{i}\, p_k\, V + 2\,\mathrm{i}\, X_a^2\, p_k\, V^3 + 2\,\mathrm{i}\, V X_a\, X_k\, p_a\, V^2\right)}{2\, m_e} \tag{37}$$

Sort the products in order to use the identities

> **(9)**, $V(X)^2\cdot$ **(9)**

$$V^3 X_l^2 = V,\; V^5 X_l^2 = V^3 \tag{38}$$

> *SortProducts*(**(37)**, [$V(X)^5$, $V(X)^3$, $X[a]^2$])

$$\left[H, Z_k\right]_- = -\frac{\left(\mathrm{i}\, p_k\, V - \mathrm{i}\, V^3 X_a^2\, p_k + \mathrm{i}\, X_a\, X_k\, p_a\, V^3 - \mathrm{i}\, V X_a\, X_k\, p_a\, V^2 - \frac{\hbar\, V^3 X_k}{2} + \frac{5\,\hbar\, V^5 X_a^2\, X_k}{2}\right)\hbar\,\kappa}{m_e} \tag{39}$$

> *SubstituteTensor*(**(38)**, **(39)**)

$$\left[H, Z_k\right]_- = -\frac{\left(\mathrm{i}\, p_k\, V - \mathrm{i}\, V p_k + \mathrm{i}\, X_a\, X_k\, p_a\, V^3 - \mathrm{i}\, V X_a\, X_k\, p_a\, V^2 + 2\,\hbar\, V^3 X_k\right)\hbar\,\kappa}{m_e} \tag{40}$$

> *Simplify*(**(40)**)

$$\left[H, Z_k\right]_- = -\frac{\hbar^2\,\kappa\left(V^3 X_k - V^5 X_a^2\, X_k\right)}{m_e} \tag{41}$$

> *SubstituteTensor*(**(38)**, **(41)**)

$$\left[H, Z_k\right]_- = 0 \tag{42}$$

And this is the result we wanted to prove.

3.2 $\left[H, Z_n\right]_- = 0$, alternative derivation using differential operators

As done in the previous section when deriving the commutators between linear and angular momentum, on the one hand, and the central potential V on the other hand, the idea here is again to use differential operators taking advantage of the ability to compute with them as operands of a product, that get applied only when it appears convenient for us

> $Setup(differentialoperators = \{[p[k], [x, y, z]]\})$

$$\left[differentialoperators = \left\{\left[p_k, [X]\right]\right\}\right] \tag{43}$$

So take the starting point **(37)**

> **(37)**

$$\left[H, Z_k\right]_- = \frac{\kappa \hbar \left(\hbar V^5 X_a^2 X_k + \hbar V^3 X_k - 2 \mathrm{i} X_a X_k p_a V^3 - 2 \mathrm{i} p_k V + 2 \mathrm{i} X_a^2 p_k V^3 + 2 \mathrm{i} V X_a X_k p_a V^2\right)}{2 m_e} \tag{44}$$

and to show that the right-hand side is equal to 0, multiply by a generic function $G(X)$ followed by transforming the products involving p_n into the application of this differential operator $p_n = -i \hbar \partial_n$

$$p_n = -\mathrm{i} \hbar \partial_n \tag{45}$$

> **(37)** $\cdot G(X)$

$$\left[H, Z_k\right]_- G(X) = \frac{\kappa \hbar \left(\hbar V^5 X_a^2 X_k + \hbar V^3 X_k - 2 \mathrm{i} X_a X_k p_a V^3 - 2 \mathrm{i} p_k V + 2 \mathrm{i} X_a^2 p_k V^3 + 2 \mathrm{i} V X_a X_k p_a V^2\right) G(X)}{2 m_e} \tag{46}$$

> $ApplyProductsOfDifferentialOperators($**(46)**$)$

$$\left[H, Z_k\right]_- G(X) = \frac{1}{2 m_e} \Bigg(\kappa \hbar \Bigg(2 \hbar V X_a X_k \left(\left(\partial_a(V) V + V \partial_a(V)\right) G(X) + V^2 \partial_a(G(X))\right) + 2 \hbar X_a^2 \Big(\big(\partial_k(V) V^2 + V \partial_k(V) V + V^2 \partial_k(V)\big) G(X) + V^3 \partial_k(G(X))\Big) - 2 \hbar V \partial_k(G(X)) - 2 \hbar \partial_k(V) G(X) - 2 \hbar X_a X_k \left(\left(\partial_a(V) V^2 + V \partial_a(V) V + V^2 \partial_a(V)\right) G(X) + V^3 \partial_a(G(X))\right) + \hbar V^5 X_a^2 X_k G(X) + \hbar V^3 X_k G(X)\Bigg)\Bigg) \tag{47}$$

> $Simplify($**(47)**$)$

$$\left[H, Z_k\right]_- G(X) = -\frac{1}{m_e} \Bigg(\hbar^2 \kappa \Bigg(-X_a^2 \partial_k(V) G(X) V^2 - V X_a^2 \partial_k(V) G(X) V - V^2 X_a^2 \partial_k(V) G(X) - V^3 X_a^2 \partial_k(G(X)) + V \partial_k(G(X)) + \partial_k(V) G(X) + X_a X_k \partial_a(V) G(X) V^2 - \frac{G(X) V^5 X_a^2 X_k}{2} - \frac{G(X) V^3 X_k}{2}\Bigg)\Bigg) \tag{48}$$

In addition, consider the application of p_l to the test function $G(X)$

> $p[l]\, G(X)$

$$p_l G(X) \tag{49}$$

> **(49)** $= ApplyProductsOfDifferentialOperators($**(49)**$)$

$$p_l G(X) = -\mathrm{i} \hbar \partial_l(G(X)) \tag{50}$$

> $isolate\left(\textbf{(50)}, \partial_l(G(X))\right)$

$$\partial_l(G(X)) = \frac{\mathrm{i} p_l G(X)}{\hbar} \tag{51}$$

Using this identity **(51)** together with the derived identity **(8)** $\equiv \partial_n(V(X)) = - V^3 X_n$, followed by multiplying by $G(X)^{-1}$

to remove the test function from the equation, we get

> $Simplify(SubstituteTensor((8), (51), (48)) \cdot G(X)^{-1})$

$$[H, Z_k]_- = -\frac{\hbar \kappa \left(\mathrm{i}\, V p_k - \mathrm{i}\, V^3 X_a^2 p_k - \frac{3 \hbar V^3 X_k}{2} + \frac{3 \hbar V^5 X_a^2 X_k}{2} \right)}{m_e} \quad \mathbf{(52)}$$

Applying **(38)** ≡ $V^3 X_l^2 = V, \;\; V^5 X_l^2 = V^3$

> $SubstituteTensor((38), (52))$

$$[H, Z_k]_- = 0 \quad \mathbf{(53)}$$

Add to the setup these derived commutation rules between the Hamiltonian, angular momentum and Runge-Lenz tensors

> **(30)**, **(53)**

$$[H, L_q]_- = 0, [H, Z_k]_- = 0 \quad \mathbf{(54)}$$

> $Setup((54))$

$$\left[algebrarules = \left\{ [H, L_q]_- = 0, [H, Z_k]_- = 0, [L_j, L_k]_- = \mathrm{i}\, \hbar\, \epsilon_{j,k,n} L_n, [L_q, V]_- = 0, [p_j, L_k]_- = \mathrm{i}\, \hbar\, \epsilon_{j,k,n} p_n, [p_k, p_l]_- = 0, [p_q, V]_- = \mathrm{i}\, \hbar\, V^3 X_q, [p_q, V^3]_- = 3\, \mathrm{i}\, \hbar\, V^5 X_q, [X_j, L_k]_- = \mathrm{i}\, \hbar\, \epsilon_{j,k,n} X_n, [X_k, p_l]_- = \mathrm{i}\, \hbar\, g_{k,l}, [X_k, V]_- = 0 \right\} \right] \quad \mathbf{(55)}$$

Reset *differentialoperators* in order to proceed to the next section working without them

> $Setup(differentialoperators = none)$

$$[differentialoperators = none] \quad \mathbf{(56)}$$

4. Commutation rules between the angular momentum L_q and the Runge-Lenz Z_k tensors

Departing from the definition of these tensors, introduced in the previous sections

> **(10)**; **(36)**

$$L_q = \epsilon_{m,n,q} X_m p_n$$

$$Z_k = \frac{-\mathrm{i}\, \hbar\, p_k + \kappa V X_k m_e + X_m p_k p_m - X_k p_m^2}{m_e} \quad \mathbf{(57)}$$

the left-hand side of the identity to be proved is the left-hand side of the commutator of these two equations

> $Commutator((10), (36))$

$$[L_q, Z_k]_- = \frac{1}{m_e} \left(\epsilon_{m,n,q}\, \hbar \left(\mathrm{i}\, X_m \left(-g_{k,n} V + V^3 X_n X_k \right) \kappa m_e + \hbar g_{k,m} p_n - 2\, \mathrm{i}\, X_k p_a p_n g_{a,m} + \mathrm{i}\, X_m p_a^2 g_{k,n} - \mathrm{i}\, X_m p_k p_a g_{a,n} + \mathrm{i}\, X_a \left(g_{a,m} p_k + g_{k,m} p_a \right) p_n \right) \right) \quad \mathbf{(58)}$$

> $Simplify((58))$

$$[L_q, Z_k]_- = -\frac{\hbar \left(\mathrm{i}\, V X_a \kappa m_e + \hbar p_a + \mathrm{i}\, X_m p_a p_m - \mathrm{i}\, X_a p_m^2 \right) \epsilon_{a,k,q}}{m_e} \quad \mathbf{(59)}$$

By eye, the right-hand side of **(59)** is similar to the right-hand side of the definition of Z_k in **(57)**, so introduce this definition directly into the right-hand side of **(59)**. For that purpose, isolate $X_k p_m^2$

> $isolate((57), X[k] \cdot p[m]^2)$

$$X_k p_m^2 = -Z_k m_e - \mathrm{i}\, \hbar\, p_k + \kappa V X_k m_e + X_m p_k p_m \quad \mathbf{(60)}$$

> $SubstituteTensor((60), (59))$

$$[L_q, Z_k]_- = \frac{i \hbar \left(-Z_a m_e + X_b p_a p_b - X_m p_a p_m\right) \epsilon_{a,k,q}}{m_e} \tag{61}$$

Simplifying, we get the desired result, and we substitute the *active* by the *inert* form of *Commutator* for posterior use of this formula without having the *Commutator* automatically executed.

> $subs(Commutator = \%Commutator, Simplify((61)))$

$$[L_q, Z_k]_- = -i \hbar Z_a \epsilon_{a,k,q} \tag{62}$$

Set now this algebra rule to be available to the system when convenient

> $Setup((62))$

$$\left[algebrarules = \left\{[H, L_q]_- = 0, [H, Z_k]_- = 0, [L_j, L_k]_- = i \hbar \epsilon_{j,k,n} L_n, [L_q, Z_k]_- = -i \hbar Z_a \epsilon_{a,k,q}, [L_q, V]_- = 0, [p_j, L_k]_- = i \hbar \epsilon_{j,k,n} p_n, [p_k, p_l]_- = 0, [p_q, V]_- = i \hbar V^3 X_q, [p_q, V^3]_- = 3 i \hbar V^5 X_q, [X_j, L_k]_- = i \hbar \epsilon_{j,k,n} X_n, [X_k, p_l]_- = i \hbar g_{k,l}, [X_k, V]_- = 0\right\}\right] \tag{63}$$

4.1 $\vec{L} \cdot \vec{Z} = \vec{Z} \cdot \vec{L} = 0$

Classically, the orbital momentum is perpendicular to the elliptic plane of motion, while the Runge-Lenz vector lies in that plane, so that $\vec{L}_{Classical} \cdot \vec{Z}_{Classical} = 0$. In quantum mechanics, from **(62)** $\equiv [L_q, Z_k]_- \neq 0$ but $\vec{L} \cdot \vec{Z} = \vec{Z} \cdot \vec{L} = 0$ still holds. To verify that, take the definition **(31)** of the quantum Runge-Lenz vector and multiply it by L_k

> $(31) . L[k]$

$$L_k Z_k = -\frac{\epsilon_{a,b,k} \left(-L_a p_b L_k + p_a L_b L_k\right)}{2 m_e} + \kappa V X_k L_k \tag{64}$$

> $Simplify((64))$

$$L_k Z_k = \kappa L_a V X_a \tag{65}$$

Using **(10)** $\equiv L_q = \epsilon_{m,n,q} X_m p_n$,

> $lhs((65)) = SubstituteTensor((10), rhs((65)))$

$$L_k Z_k = \kappa \epsilon_{a,m,n} X_m p_n V X_a \tag{66}$$

> $Simplify((66))$

$$L_k Z_k = 0 \tag{67}$$

and due to **(62)** $\equiv [L_q, Z_k]_- = -\hbar i Z_a \epsilon_{a,k,q}$, reversing the order in the product,

> $SortProducts((67), [Z[k], L[k]])$

$$Z_k L_k = 0 \tag{68}$$

5. Commutation rules between the components of the Runge-Lenz tensor

Here again the starting point is **(36)**, the definition of the quantum Runge-Lenz tensor

> $SubstituteTensorIndices(k = q, (36))$

$$Z_q = \frac{-i \hbar p_q + \kappa V X_q m_e + X_m p_q p_m - X_q p_m^2}{m_e} \tag{69}$$

The commutator $[Z_k, Z_q]_-$ is computed via

> $Commutator((36), (69))$

$$\left[Z_k, Z_q\right]_- = \frac{1}{m_e^2}\Big(-2\,\mathrm{i}\,\hbar\, g_{m,q} X_k p_m p_a^2 + \kappa\, m_e X_m \left(\mathrm{i}\,\hbar\, p_k \left(-g_{m,q} V + V^3 X_m X_q\right) + \mathrm{i}\,\hbar \left(-g_{k,q} V + V^3 X_k X_q\right) p_m\right) + 2\,\mathrm{i}\,\hbar\, g_{a,m} X_k p_m p_q p_a - 2\,\mathrm{i}\,\hbar\, V X_q p_a \kappa\, m_e g_{a,k} + \hbar^2 V^3 X_k X_q \kappa\, m_e - \hbar^2 V^3 X_q X_k \kappa\, m_e - \mathrm{i}\,\hbar\, X_m \left(g_{a,k} p_q p_a p_m + g_{a,m} p_k p_q p_a\right) + 2\,\mathrm{i}\,\hbar\, g_{a,k} X_q p_a p_m^2 + \mathrm{i}\,\hbar\, X_a \left(g_{a,m} p_q + g_{m,q} p_a\right) p_k p_m + \mathrm{i}\,\hbar\, X_q \left(2 p_a V^3 X_a + V^3 \left(3\,\mathrm{i}\,\hbar + 2 p_a X_a\right) - \left(3\,\mathrm{i}\,\hbar\, V^5 X_a + 2 V^3 p_a\right) X_a\right) X_k \kappa\, m_e - 2\,\mathrm{i}\,\hbar\, g_{a,m} X_q p_a p_k p_m - \hbar^2 g_{k,q} p_m^2 + \hbar^2 g_{m,q} p_k p_m - \hbar^2 p_q p_a g_{a,k} + \hbar^2 p_a^2 g_{k,q} + \Bigg(\Bigg(2\,\mathrm{i}\, X_k \left(V p_m g_{m,q} - \frac{\left(2 p_m V^3 X_m + V^3 \left(3\,\mathrm{i}\,\hbar + 2 p_m X_m\right) - \left(3\,\mathrm{i}\,\hbar\, V^5 X_m + 2 V^3 p_m\right) X_m\right) X_q}{2}\right) - \mathrm{i}\, X_a \left(p_q V^3 X_a + V^3 X_q p_a\right) X_k + \mathrm{i}\, V X_a \left(g_{a,k} p_q + g_{k,q} p_a\right)\Bigg) \kappa\, m_e + \mathrm{i}\, X_m \left(g_{k,q} p_a^2 p_m + g_{m,q} p_k p_a^2\right) - \mathrm{i}\, X_a \left(g_{a,k} p_q + g_{k,q} p_a\right) p_m^2\Bigg) \hbar\Big) \tag{70}$$

> $Simplify(\textbf{(70)})$

$$\left[Z_k, Z_q\right]_- = \frac{1}{m_e^2}\left(\mathrm{i}\,\hbar \left(\kappa X_a^2 X_q p_k V^3 m_e + 3\,\kappa X_k p_q V m_e - 3\,\kappa X_q p_k V m_e - X_a^2 X_k p_q V^3 \kappa\, m_e + g_{k,q} X_m p_a^2 p_m - g_{k,q} X_a p_m^2 p_a + X_q p_a^2 p_k - X_k p_a^2 p_q\right)\right) \tag{71}$$

In order to use **(9)** ≡ $V^3 X_l^2 = V$, sort the products in **(71)** using the ordering $V^3 X_a^2$

> $Normal(SortProducts(\textbf{(71)}, [V(X)^3, X[a]^2]))$

$$\left[Z_k, Z_q\right]_- = \frac{1}{m_e^2}\left(\mathrm{i}\,\hbar \left(\kappa V^3 X_a^2 X_q p_k m_e - V^3 X_a^2 X_k p_q \kappa\, m_e + 3\,\kappa X_k p_q V m_e - 3\,\kappa X_q p_k V m_e + g_{k,q} X_m p_a^2 p_m - g_{k,q} X_a p_m^2 p_a + X_q p_a^2 p_k - X_k p_a^2 p_q\right)\right) \tag{72}$$

> $SubstituteTensor(\textbf{(9)}, \textbf{(72)})$

$$\left[Z_k, Z_q\right]_- = \frac{1}{m_e^2}\left(\mathrm{i}\,\hbar \left(\kappa\, m_e V X_q p_k - V X_k p_q \kappa\, m_e + 3\,\kappa X_k p_q V m_e - 3\,\kappa X_q p_k V m_e + g_{k,q} X_m p_a^2 p_m - g_{k,q} X_a p_m^2 p_a + X_q p_a^2 p_k - X_k p_a^2 p_q\right)\right) \tag{73}$$

Regarding the term quadratic in the momentum, from the expression for the Hamiltonian **(5)** ≡ $H = \frac{p_l^2}{2\ m_e} - \kappa V$,

> $isolate(\textbf{(5)}, p[l]^2)$

$$p_l^2 = 2\left(\kappa V + H\right) m_e \tag{74}$$

In order to use this equation **(74)** to substitute p_l^2 into the expression **(73)** for $\left[Z_k, Z_q\right]_-$ and *not* receive noncommutative products with H in between the position X_k and momentum p_q tensors (that would require using afterwards the commutator between H and p_q), sort first the products in **(73)** positioning all square of momentums p^2 to the right of occurrences of p

> $SortProducts(\textbf{(73)}, [p[a], p[k], p[m], p[q], p[a]^2, p[m]^2])$

(75)

$$\left[Z_k, Z_q\right]_- = \frac{1}{m_e^2}\left(-\mathrm{i}\,\hbar\left(-\kappa\, m_e\, V X_q\, p_k + V X_k\, p_q\, \kappa\, m_e - 3\,\kappa X_k\, p_q\, V m_e + 3\,\kappa X_q\, p_k\, V m_e - g_{k,q}\, X_m\, p_m\, p_a^2 + g_{k,q}\, X_a\, p_a\, p_m^2 + X_k\, p_q\, p_a^2 - X_q\, p_k\, p_a^2\right)\right) \tag{75}$$

> *SubstituteTensor*(**(74)**, **(75)**)

$$\left[Z_k, Z_q\right]_- = \frac{1}{m_e^2}\left(\mathrm{i}\,\hbar\left(\kappa\, m_e\, V X_q\, p_k - V X_k\, p_q\, \kappa\, m_e + 3\,\kappa X_k\, p_q\, V m_e - 3\,\kappa X_q\, p_k\, V m_e + g_{k,q}\, X_m\, p_m\, 2\left(\kappa V + H\right) m_e - g_{k,q}\, X_a\, p_a\, 2\left(\kappa V + H\right) m_e - X_k\, p_q\, 2\left(\kappa V + H\right) m_e + X_q\, p_k\, 2\left(\kappa V + H\right) m_e\right)\right) \tag{76}$$

> *Simplify*(**(76)**)

$$\left[Z_k, Z_q\right]_- = \frac{2\,\mathrm{i}\,\hbar\left(-X_k\, p_q\, H + X_q\, p_k\, H\right)}{m_e} \tag{77}$$

Finally, from the definition of the angular momentum **(10)** ≡ $L_q = \epsilon_{m,n,q}\, X_m\, p_n$, multiplying by $\epsilon_{a,b,c}$ we can construct an expression for $X_a\, p_b\, H - X_b\, p_a\, H$ in terms of L_q

> *LeviCivita*[a, b, q] · **(10)**

$$\epsilon_{a,b,q}\, L_q = \epsilon_{a,b,q}\, \epsilon_{m,n,q}\, X_m\, p_n \tag{78}$$

> *Simplify*((*rhs* = *lhs*)(**(78)**))

$$X_a\, p_b - X_b\, p_a = \epsilon_{a,b,q}\, L_q \tag{79}$$

> *Expand*(**(79)** · *H*)

$$X_a\, p_b\, H - X_b\, p_a\, H = \epsilon_{a,b,q}\, H\, L_q \tag{80}$$

> *SubstituteTensor*(**(80)**, **(77)**)

$$\left[Z_k, Z_q\right]_- = \frac{-2\,\mathrm{i}\,\hbar\,\epsilon_{c,k,q}\, H\, L_c}{m_e} \tag{81}$$

Which is the identity we wanted to prove.

5.1 Alternative demonstration using differential operators

Set again the *differentialoperator* representation for the momentum operator p_k

> *Setup*(*differentialoperators* = {[p[k], [x, y, z]]})

$$\left[\mathit{differentialoperators} = \left\{\left[p_k, [X]\right]\right\}\right] \tag{82}$$

and apply the expression **(70)** for $\left[Z_k, Z_q\right]_-$ to the test function $G(X)$

> *ApplyProductsOfDifferentialOperators*(**(70)** · *G*(*X*))

$$\left[Z_k, Z_q\right]_- G(X) = -\frac{1}{m_e^2}\left(\hbar\left(-\hbar^3 X_m\left(\partial_a\left(\partial_m\left(\partial_q\left(G(X)\right)\right)\right) g_{a,k} + \partial_a\left(\partial_k\left(\partial_q\left(G(X)\right)\right)\right) g_{a,m}\right) - \hbar\, V^3 X_k\, X_q\, \kappa\, m_e\, G(X) + \hbar\, V^3 X_q\, X_k\, \kappa\, m_e\, G(X) + 2\, g_{a,k}\, \hbar^3 X_q\, \partial_a\left(\Box\left(G(X)\right)\right) - 2\, g_{m,q}\, \hbar^3 X_k\, \partial_m\left(\Box\left(G(X)\right)\right) + 2\, g_{a,m}\, \hbar^3 X_k\, \partial_a\left(\partial_m\left(\partial_q\left(G(X)\right)\right)\right) + \hbar^3 X_a\left(\partial_k\left(\partial_m\left(\partial_q\left(G(X)\right)\right)\right) g_{a,m} + \partial_a\left(\partial_k\left(\partial_m\left(G(X)\right)\right)\right) g_{m,q}\right) - 2\, g_{a,m}\, \hbar^3 X_q\, \partial_a\left(\partial_k\left(\partial_m\left(G(X)\right)\right)\right) + g_{m,q}\, \hbar^3\, \partial_k\left(\partial_m\left(G(X)\right)\right) - g_{a,k}\, \hbar^3\, \partial_a\left(\partial_q\left(G(X)\right)\right) + \hbar\,\kappa\, m_e\, X_q\left(-2\left(\partial_a(V)\, V^2 + V\,\partial_a(V)\, V + V^2\,\partial_a(V)\right) X_a\, X_k\, G(X) - 8\, V^3 X_k\, G(X) - 2\, V^3 X_a\, X_k\, \partial_a\left(G(X)\right) - V^3\left(2\, X_a\, X_k\, \partial_a\left(G(X)\right) + 5\, X_k\, G(X)\right) - 3\, V^5 X_a^2\, X_k\, G(X) + 2\, V^3\left(4\, X_k\, G(X) + X_a\, X_k\, \partial_a\left(G(X)\right)\right)\right) + \hbar\,\kappa\, m_e\, X_k\left(-2\, g_{m,q}\, V\,\partial_m\left(G(X)\right) + V^3\left(2\, X_m\, X_q\, \partial_m\left(G(X)\right)\right.\right.\right.\right. \tag{83}$$

$$+ 5 X_q G(X)\Big) + 8 V^3 X_q G(X) + 2 \left(\partial_m(V) V^2 + V \partial_m(V) V + V^2 \partial_m(V)\right) X_m X_q G(X)$$
$$+ 2 V^3 X_m X_q \partial_m(G(X)) + 3 V^5 X_m^2 X_q G(X) - 2 V^3 \left(4 X_q G(X) + X_m X_q \partial_m(G(X))\right)\Big)$$
$$+ 2 \kappa m_e g_{a,k} \hbar V X_q \partial_a(G(X)) - \hbar \kappa m_e V X_a \left(\partial_q(G(X)) g_{a,k} + \partial_a(G(X)) g_{k,q}\right)$$
$$+ \hbar^3 X_m \left(\partial_m(\Box(G(X))) g_{k,q} + \partial_k(\Box(G(X))) g_{m,q}\right) - \hbar^3 X_a \left(\partial_q(\Box(G(X))) g_{a,k} + \partial_a(\Box(G(X))) g_{k,q}\right)$$
$$+ \hbar \kappa m_e X_a \Big(g_{a,q} V^3 X_k G(X) + g_{k,q} V^3 X_a G(X) + \left(\partial_q(V) V^2 + V \partial_q(V) V + V^2 \partial_q(V)\right) X_a X_k G(X)$$
$$+ V^3 X_a X_k \partial_q(G(X)) + V^3 X_q \left(g_{a,k} G(X) + X_k \partial_a(G(X))\right)\Big) + \kappa m_e \hbar X_m \Big(g_{m,q} V \partial_k(G(X))$$
$$+ g_{m,q} \partial_k(V) G(X) - g_{k,m} V^3 X_q G(X) - g_{k,q} V^3 X_m G(X) - V^3 X_m X_q \partial_k(G(X)) - \Big(\partial_k(V) V^2$$
$$+ V \partial_k(V) V + V^2 \partial_k(V)\Big) X_m X_q G(X) - \left(-g_{k,q} V + V^3 X_k X_q\right) \partial_m(G(X))\Big)\Big)$$

> *Simplify*(**(83)**)

$$[Z_k, Z_q]_- G(X) = \frac{1}{m_e^2} \Big(\hbar^2 \Big(\hbar^2 X_k \partial_q(\Box(G(X))) - \hbar^2 X_q \partial_k(\Box(G(X))) + 3 \kappa V X_k \partial_q(G(X)) m_e$$
$$- 3 \kappa V X_q \partial_k(G(X)) m_e - \kappa X_q \partial_k(V) G(X) m_e - \kappa G(X) V^3 X_k X_q m_e - \kappa V^3 X_d^2 X_k \partial_q(G(X)) m_e + \kappa V^3$$
$$X_d^2 X_q \partial_k(G(X)) m_e + \kappa V^2 X_d \partial_k(V) G(X) X_d X_q m_e - \kappa V^2 X_d \partial_q(V) G(X) X_d X_k m_e$$
$$- 2 \kappa V^2 X_k \partial_d(V) G(X) X_d X_q m_e + 2 \kappa V^2 X_q \partial_d(V) G(X) X_d X_k m_e + \kappa V X_d \partial_k(V) G(X) V X_d X_q m_e$$
$$- \kappa V X_d \partial_q(V) G(X) V X_d X_k m_e - 2 \kappa V X_k \partial_d(V) G(X) V X_d X_q m_e + 2 \kappa V X_q \partial_d(V) G(X) V X_d X_k m_e$$
$$+ \kappa X_d \partial_k(V) G(X) V^2 X_d X_q m_e - \kappa X_d \partial_q(V) G(X) V^2 X_d X_k m_e - 2 \kappa X_k \partial_d(V) G(X) V^2 X_d X_q m_e$$
$$+ 2 \kappa X_q \partial_d(V) G(X) V^2 X_d X_k m_e\Big)\Big) \quad \mathbf{(84)}$$

Recalling **(8)** $\equiv \partial_n(V) = -V^3 X_n$ and **(51)** $\equiv \partial_l(G) = \frac{\mathrm{i}}{\hbar} p_l G$

> *Simplify*(*SubstituteTensor*(**(8)**, **(51)**, **(84)**))

$$[Z_k, Z_q]_- G(X) = \frac{1}{m_e^2} \Bigg(3 \hbar \Bigg(-\frac{\mathrm{i} V^3 X_d^2 X_k p_q G(X) \kappa m_e}{3} + \frac{\mathrm{i} V^3 X_d^2 X_q p_k G(X) \kappa m_e}{3} + \frac{\hbar^3 X_k \partial_q(\Box(G(X)))}{3}$$
$$- \frac{\hbar^3 X_q \partial_k(\Box(G(X)))}{3} + \mathrm{i} \left(V X_k p_q G(X) - V X_q p_k G(X)\right) m_e \kappa\Bigg)\Bigg) \quad \mathbf{(85)}$$

Evaluating the term $\partial_q(\Box(G(X)))$

> $p[a]\, p[l]^2 \cdot G(X)$

$$p_l^2 p_a G(X) \quad \mathbf{(86)}$$

> **(86)** = *ApplyProductsOfDifferentialOperators*(**(86)**)

$$p_l^2 p_a G(X) = \mathrm{i} \hbar^3 \partial_a(\Box(G(X))) \quad \mathbf{(87)}$$

> *isolate*(**(87)**, $\partial_a(\Box(G(X)))$)

$$\partial_a(\Box(G(X))) = \frac{-\mathrm{i} p_l^2 p_a G(X)}{\hbar^3} \quad \mathbf{(88)}$$

Inserting this result into the expression **(85)** for $[Z_k, Z_q]_-$ and removing the test function multiplying by $G(X)^{-1}$

> *Simplify*(*SubstituteTensor*(**(88)**, **(85)**) $\cdot G(X)^{-1}$)

(89)

$$[Z_k, Z_q]_- = \frac{i\hbar\left(3 V X_k p_q \kappa m_e - \kappa m_e V^3 X_d^2 X_k p_q + \kappa m_e V^3 X_d^2 X_q p_k - 3\kappa m_e V X_q p_k + X_q p_d^2 p_k - X_k p_d^2 p_q\right)}{m_e^2} \tag{89}$$

This expression can be factored

> *Factor*((**89**))

$$[Z_k, Z_q]_- = \frac{-i\hbar\left(-3 V \kappa m_e + p_d^2 + \kappa m_e V^3 X_d^2\right)\left(-X_q p_k + X_k p_q\right)}{m_e^2} \tag{90}$$

Using the identity **(9)** $\equiv V^3 X_l^2 = V$ for the potential

> *SubstituteTensor*((**9**), (**90**))

$$[Z_k, Z_q]_- = \frac{-i\hbar\left(-2 V \kappa m_e + p_d^2\right)\left(-X_q p_k + X_k p_q\right)}{m_e^2} \tag{91}$$

Next using

> (**74**), (**79**)

$$p_l^2 = 2(\kappa V + H) m_e, X_a p_b - X_b p_a = \epsilon_{a,b,q} L_q \tag{92}$$

> *SubstituteTensor*((**92**), (**91**))

$$[Z_k, Z_q]_- = \frac{-2 i\hbar \epsilon_{c,k,q} H L_c}{m_e} \tag{93}$$

Which is the expected result. Set now differential operators to *none*.

> *Setup*(*differentialoperators* = *none*)

$$[differentialoperators = none] \tag{94}$$

6. The square of the norm of the Runge-Lenz vector

Taking the square of the definition of Z_k and simplifying

> (**31**)^2

$$Z_k^2 = \left(\frac{\epsilon_{a,b,k}(L_a p_b - p_a L_b)}{2 m_e} + \kappa V X_k\right)\left(\frac{\epsilon_{c,d,k}(L_c p_d - p_c L_d)}{2 m_e} + \kappa V X_k\right) \tag{95}$$

> *Simplify*((**95**))

$$Z_k^2 = \frac{1}{2 m_e^2}\left(2\hbar^2 \kappa V^3 X_a^2 m_e - 6\kappa\hbar^2 V m_e + 2\kappa^2 V^2 X_a^2 m_e^2 - 4\epsilon_{a,b,c} X_a p_b L_c V \kappa m_e - p_a p_b L_b L_a + 2 p_a^2 L_b^2 - p_a L_b L_a p_b\right) \tag{96}$$

Using the algebraic properties of the potential

> (**9**), $V(X)^{-1}\cdot$(**9**)

$$V^3 X_l^2 = V, V^2 X_l^2 = 1 \tag{97}$$

the expression **(96)** for Z_k^2 becomes

> *SubstituteTensor*((**97**), (**96**))

$$Z_k^2 = \frac{-4\kappa\hbar^2 V m_e + 2\kappa^2 m_e^2 - 4\epsilon_{a,b,c} X_a p_b L_c V \kappa m_e - p_a p_b L_b L_a + 2 p_a^2 L_b^2 - p_a L_b L_a p_b}{2 m_e^2} \tag{98}$$

The term having $\epsilon_{a,b,c}$ can be simplified using the expression of the momentum operator

> $(rhs = lhs)((\mathbf{10}))$

$$\epsilon_{m,n,q} X_m p_n = L_q \qquad \mathbf{(99)}$$

> $\mathbf{(99)} \cdot L[q] \cdot V(X)$

$$\epsilon_{m,n,q} X_m p_n L_q V = L_q^2 V \qquad \mathbf{(100)}$$

> $SubstituteTensor(\mathbf{(100)}, \mathbf{(98)})$

$$Z_k^2 = \frac{-4 \kappa \hbar^2 V m_e + 2 \kappa^2 m_e^2 - 4 L_c^2 V \kappa m_e - p_a p_b L_b L_a + 2 p_a^2 L_b^2 - p_a L_b L_a p_b}{2 m_e^2} \qquad \mathbf{(101)}$$

Reordering **(101)** to have the two terms with four operators sorted as $p_a p_b L_a L_b$

> $Simplify(SortProducts(\mathbf{(101)}, [p[a], p[b], L[a], L[b]]))$

$$Z_k^2 = \frac{-2 \kappa \hbar^2 V m_e + \kappa^2 m_e^2 + \hbar^2 p_a^2 - 2 L_a^2 V \kappa m_e + p_a^2 L_b^2 - p_a p_b L_a L_b}{m_e^2} \qquad \mathbf{(102)}$$

Considering now the resulting single term $p_a p_b L_a L_b$, it can be shown it is equal to zero using the definition **(10)** ≡ $L_q = \epsilon_{m,n,q} X_m p_n$

> $p[a]\, p[b]\, L[a]\, L[b]$

$$p_a p_b L_a L_b \qquad \mathbf{(103)}$$

> $\mathbf{(103)} = SubstituteTensor(\mathbf{(10)}, \mathbf{(103)})$

$$p_a p_b L_a L_b = \epsilon_{a,m,n} \epsilon_{b,e,f} p_a p_b X_m p_n X_e p_f \qquad \mathbf{(104)}$$

> $Simplify(\mathbf{(104)})$

$$p_a p_b L_a L_b = 0 \qquad \mathbf{(105)}$$

Taking this result into account, we have, for Z_k^2

> $subs(\mathbf{(105)}, \mathbf{(102)})$

$$Z_k^2 = \frac{-2 \kappa \hbar^2 V m_e + \kappa^2 m_e^2 + \hbar^2 p_a^2 - 2 L_a^2 V \kappa m_e + p_a^2 L_b^2}{m_e^2} \qquad \mathbf{(106)}$$

Substituting now **(74)** ≡ $p_l^2 = 2 (\kappa V + H) m_e$

> $SubstituteTensor(\mathbf{(74)}, \mathbf{(106)})$

$$Z_k^2 = \frac{2 \hbar^2 H m_e + \kappa^2 m_e^2 - 2 L_a^2 V \kappa m_e + 2 (\kappa V + H) m_e L_b^2}{m_e^2} \qquad \mathbf{(107)}$$

Equalizing the repeated indices, the right-hand side can be factored, resulting in

> $(lhs = Factor@rhs)(EqualizeRepeatedIndices(\mathbf{(107)}) - \kappa^2) + \kappa^2$

$$Z_k^2 = \frac{2 H (\hbar^2 + L_a^2)}{m_e} + \kappa^2 \qquad \mathbf{(108)}$$

Which is the result we wanted to demonstrate.

7. The atomic hydrogen spectrum

We now have all the algebra to reconstruct the hydrogen spectrum. Following the literature, this approach is limited to the bound states for which the energy is negative. Assuming an eigenstate of H with negative eigenvalue E, we now replace the Hamiltonian H by E, and look for the possible values of E. Another way to state the same thing is that the analysis is restricted to the subspace of energy E. The operator $M_n = \sqrt{-\frac{m_e}{2 E}}\, Z_n$, is introduced as mentioned in the presentation. The operators J and K, to be used soon after, are added to the system.

> $Setup(hermitianoperators = \{M, J, K\})$

$$[hermitianoperators = \{H, J, K, L, M, V, p, x, y, z\}] \tag{109}$$

> $Define(M[n], J[n], K[n], quiet)$

$$\{\gamma_a, J_n, K_n, L_k, M_n, \sigma_a, Z_k, \partial_a, g_{a,b}, p_k, \epsilon_{a,b,c}, X_a\} \tag{110}$$

> $Assume(m_e > 0, E < 0)$

$$\{E::(-\infty, 0)\}, \{m_e::(0, \infty)\} \tag{111}$$

> $M[n] = \sqrt{-\frac{m_e}{2E}}\, Z[n]$

$$M_n = \frac{\sqrt{-\frac{2 m_e}{E}}\, Z_n}{2} \tag{112}$$

> $simplify(isolate(\mathbf{(112)}, Z[n]))$

$$Z_n = \frac{M_n \sqrt{2}\,\sqrt{-E}}{\sqrt{m_e}} \tag{113}$$

Recalling the commutation rules **(93)** $\equiv [Z_k, Z_q]_- = -\frac{2\,\mathrm{i}\,\hbar\,\epsilon_{a,k,q} H L_a}{m_e}$ and **(113)** above with E replacing H

> $SubstituteTensor(H = E, \mathbf{(113)}, \mathbf{(93)})$

$$\left[\frac{M_k \sqrt{2}\,\sqrt{-E}}{\sqrt{m_e}}, \frac{M_q \sqrt{2}\,\sqrt{-E}}{\sqrt{m_e}}\right]_- = \frac{-2\,\mathrm{i}\,\hbar\,\epsilon_{c,k,q} E L_c}{m_e} \tag{114}$$

> $Simplify(\mathbf{(114)})$

$$-\frac{2E\,[M_k, M_q]_-}{m_e} = \frac{-2\,\mathrm{i}\,\hbar\,\epsilon_{c,k,q} E L_c}{m_e} \tag{115}$$

Isolating the commutator, the expression **(93)** for $[Z_k, Z_q]_-$ appears rewritten in terms of the M_k as

> $isolate(\mathbf{(115)}, Commutator(M[k], M[q]))$

$$[M_k, M_q]_- = \mathrm{i}\,\hbar\,\epsilon_{c,k,q} L_c \tag{116}$$

Likewise, inserting **(113)** $\equiv Z_n = \frac{M_n \sqrt{2}\,\sqrt{-E}}{\sqrt{m_e}}$ into the expression **(62)** $\equiv [L_q, Z_k]_- = -\hbar\,\mathrm{i}\,\epsilon_{a,k,q} Z_a$, we get it rewritten in terms of L_q, M_k

> $Simplify(SubstituteTensor(\mathbf{(113)}, \mathbf{(62)}))$

$$\frac{\sqrt{2}\,\sqrt{-E}\,[L_q, M_k]_-}{\sqrt{m_e}} = \frac{-\mathrm{i}\,\hbar\,M_a \sqrt{2}\,\sqrt{-E}\,\epsilon_{a,k,q}}{\sqrt{m_e}} \tag{117}$$

> $isolate(\mathbf{(117)}, Commutator(L[q], M[k]))$

$$[L_q, M_k]_- = -\mathrm{i}\,\hbar\,M_a\,\epsilon_{a,k,q} \tag{118}$$

Add these two newly derived commutators to the setup

> $Setup(\mathbf{(116)}, \mathbf{(118)})$

$$\left[algebrarules = \{[H, L_q]_- = 0, [H, Z_k]_- = 0, [L_j, L_k]_- = \mathrm{i}\,\hbar\,\epsilon_{j,k,n} L_n, [L_q, M_k]_- = -\mathrm{i}\,\hbar\,M_a\,\epsilon_{a,k,q}, [L_q, Z_k]_- = -\mathrm{i}\,\hbar\,Z_a\,\epsilon_{a,k,q}, [L_q, V]_- = 0, [M_k, M_q]_- = \mathrm{i}\,\hbar\,\epsilon_{c,k,q} L_c, [p_j, L_k]_- = \mathrm{i}\,\hbar\,\epsilon_{j,k,n} p_n, [p_k, p_l]_- = 0, [p_q, V]_- \right. \tag{119}$$

$= \mathrm{i}\,\hbar\, V^3 X_q,\ \left[p_q, V^3\right]_- = 3\,\mathrm{i}\,\hbar\, V^5 X_q,\ \left[X_j, L_k\right]_- = \mathrm{i}\,\hbar\, \epsilon_{j,k,n} X_n,\ \left[X_k, p_l\right]_- = \mathrm{i}\,\hbar\, g_{k,l},\ \left[X_k, V\right]_- = 0 \}]$

These commutators **(118)**, **(116)**, together with the departing commutator

> $(\%Commutator = Commutator)(L[m], L[n])$

$$\left[L_m, L_n\right]_- = \mathrm{i}\,\hbar\, \epsilon_{a,m,n} L_a \quad \textbf{(120)}$$

constitute a closed form, the algebra of the SO(4) group, that is, the rotation group in dimension 4.

We now define the two operators J and K as follows

> $J[m] = \frac{1}{2} \cdot (L[m] + M[m])$

$$J_m = \frac{L_m}{2} + \frac{M_m}{2} \quad \textbf{(121)}$$

> $K[m] = \frac{1}{2} \cdot (L[m] - M[m])$

$$K_m = \frac{L_m}{2} - \frac{M_m}{2} \quad \textbf{(122)}$$

Because M and L both commute with H (since M is proportional to Z up-to a commutative factor), it is straightforward to see that J and K commute with H. They are therefore a constant of the motion. Additionally after having set the commutators **(116)** $\equiv \left[M_k, M_q\right]_-$ and **(118)** $\equiv \left[L_q, M_k\right]_-$ derived from the results of the previous sections, the commutator between the components of J_m results in

> $Commutator(\textbf{(121)}, SubstituteTensorIndices(m = n, \textbf{(121)}))$

$$\left[J_m, J_n\right]_- = \frac{\mathrm{i}}{4}\,\hbar\left(\left(L_a + 2\,M_a\right)\epsilon_{a,m,n} + \epsilon_{c,m,n} L_c\right) \quad \textbf{(123)}$$

> $Simplify(\textbf{(123)})$

$$\left[J_m, J_n\right]_- = \frac{\mathrm{i}}{2}\,\epsilon_{a,m,n}\,\hbar\left(L_a + M_a\right) \quad \textbf{(124)}$$

> $SubstituteTensor((rhs = lhs)(\textbf{(121)}), \textbf{(124)})$

$$\left[J_m, J_n\right]_- = \mathrm{i}\,\epsilon_{a,m,n}\,\hbar\, J_a \quad \textbf{(125)}$$

In a similar manner

> $Commutator(\textbf{(122)}, SubstituteTensorIndices(m = n, \textbf{(122)}))$

$$\left[K_m, K_n\right]_- = \frac{\mathrm{i}}{4}\,\hbar\left(\left(L_a - 2\,M_a\right)\epsilon_{a,m,n} + \epsilon_{c,m,n} L_c\right) \quad \textbf{(126)}$$

> $Simplify(\textbf{(126)})$

$$\left[K_m, K_n\right]_- = \frac{\mathrm{i}}{2}\,\epsilon_{a,m,n}\,\hbar\left(L_a - M_a\right) \quad \textbf{(127)}$$

> $SubstituteTensor((rhs = lhs)(\textbf{(122)}), \textbf{(127)})$

$$\left[K_m, K_n\right]_- = \mathrm{i}\,\epsilon_{a,m,n}\,\hbar\, K_a \quad \textbf{(128)}$$

Also

> $Commutator(\textbf{(121)}, subs(m = n, \textbf{(122)}))$

$$\left[J_m, K_n\right]_- = \frac{\mathrm{i}}{4}\,\hbar\left(\epsilon_{a,m,n} L_a - \epsilon_{c,m,n} L_c\right) \quad \textbf{(129)}$$

> $Simplify(\textbf{(129)})$

$$\left[J_m, K_n\right]_- = 0 \quad \textbf{(130)}$$

Both J and K have the symmetry of a rotation operator in two independent 3 dimension spaces. H then has the symmetry of the group SO(3)⊗SO(3). Furthermore, one knows that the possible eigenvalues for the rotation operators J and K are $j\,(j+1)\,\hbar^2$ and $k\,(k+1)\,\hbar^2$, with $j, k \in \left\{0, \frac{1}{2}, 1, \frac{3}{2}, 2, \ldots\right\}$. Now, computing J^2 and K^2

> $Expand(\textbf{(121)}^2)$

$$J_m^2 = \frac{L_m^2}{4} + \frac{L_m M_m}{2} + \frac{M_m^2}{4} \quad \textbf{(131)}$$

Recalling **(67)** ≡ $L_k\, Z_k = 0$, and considering that M is proportional to Z, we have that $L_m\, M_m = 0$

> $subs(L[m]\, M[m] = 0,$ **(131)**)

$$J_m^2 = \frac{L_m^2}{4} + \frac{M_m^2}{4} \tag{132}$$

Next, from **(122)** ≡ $K_m = \frac{L_m}{2} - \frac{M_m}{2}$

> $Expand($**(122)**$^2)$

$$K_m^2 = \frac{L_m^2}{4} - \frac{L_m\, M_m}{2} + \frac{M_m^2}{4} \tag{133}$$

> $subs(L[m]\, M[m] = 0,$ **(133)**)

$$K_m^2 = \frac{L_m^2}{4} + \frac{M_m^2}{4} \tag{134}$$

So that

> **(132)**−**(134)**

$$J_m^2 - K_m^2 = 0 \tag{135}$$

That is, $J_m^{\ 2} = K_m^{\ 2}$, which means they share the same eigenvalues, say $j\,(j+1)\,\hbar^2$ for a given eigenstate of H with the considered eigenvalue E.

Next, inserting **(113)** ≡ $Z_n = \frac{M_n \sqrt{2}\,\sqrt{-E}}{\sqrt{m_e}}$ into **(108)** ≡ $Z_k^2 = \frac{2\,H\left(\hbar^2 + L_a^2\right)}{m_e} + \kappa^2$ we get an expression for M_k^2

> $SubstituteTensor(H = E,$ **(113)**, **(108)**)

$$-\frac{2\,E\,M_k^2}{m_e} = \frac{2\,E\left(\hbar^2 + L_a^2\right)}{m_e} + \kappa^2 \tag{136}$$

> $-\frac{m_e}{2\,E}$ **(136)**

$$M_k^2 = -\frac{2\,\hbar^2\,E + 2\,E\,L_a^2 + \kappa^2\,m_e}{2\,E} \tag{137}$$

Substituting this result into the expression **(132)** for J_m^2 and simplifying we get

> $Simplify(SubstituteTensor($**(137)**, **(132)**))

$$J_m^2 = -\frac{\hbar^2}{4} - \frac{\kappa^2\,m_e}{8\,E} \tag{138}$$

Taking the average value of J_m^2 over an eigenvector, J_m^2 can be replaced by its eigenvalue $j\,(j+1)\,\hbar^2$

> $subs(J[m]^2 = j\,(j+1)\,\hbar^2,$ **(138)**)

$$j\,(j+1)\,\hbar^2 = -\frac{\hbar^2}{4} - \frac{\kappa^2\,m_e}{8\,E} \tag{139}$$

from where the possible values of the energy are

> $isolate($**(139)**$, E)$

$$E = -\frac{\kappa^2\,m_e}{2\,\hbar^2\,(2\,j+1)^2} \tag{140}$$

Assuming $n = 2\,j + 1$, a positive integer and $j \in \left\{0, \frac{1}{2}, 1, \frac{3}{2}, 2, \ldots\right\}$, the spectrum for an hydrogen atom is thus

> $subs(\{2\,j + 1 = n,\ E = E(n)\},$ **(140)**)

(141)

$$E(n) = -\frac{\kappa^2 m_e}{2 \hbar^2 n^2} \qquad \textbf{(141)}$$

Which is the energy spectrum for a spinless hydrogenoid system.

Conclusions

In this presentation, we derived, step-by-step, the SO(4) symmetry of the Hydrogen atom and its spectrum using the symbolic computer algebra Maple system. The derivation was performed without departing from the results, entering only the main definition formulas in eqs. **(1)**, **(2)** and **(5)**, followed by using a few simplification commands - mainly *Simplify*, *SortProducts* and *SubstituteTensor* - and a handful of Maple basic commands, *subs*, *lhs*, *rhs* and *isolate*. The computational path that was used to get the results of sections 2 to 7 is not unique. Instead of searching for the shortest path, we prioritized clarity and illustration of the techniques that can be used to crack problems like this one.

This problem is mainly about simplifying expressions using two different techniques. First, expressions with noncommutative operands in products need reduction with respect to the commutator algebra rules that have been set. Second, products of tensorial operators require simplification using the sum rule for repeated indices and the symmetries of tensorial subexpressions. Those techniques, which are part of the Maple Physics simplifier, together with the *SortProducts* and *SubstituteTensor* commands for sorting the operands in products to apply tensorial identities, sufficed. The derivations were performed in a reasonably small number of steps.

Two different computational strategies - with and without differential operators - were used in sections 3 and 5, showing an approach for verifying results, a relevant issue in general when performing complicated algebraic manipulations. The Maple Physics ability to handle differential operators as noncommutative operands in products (as frequently done in paper and pencil computations) facilitates readability and ease in entering the computations. The complexity of those operations is then handled by one *Physics:-Library* command, *ApplyProductsOfDifferentialOperators* (see eqs. **(47)** and **(83)**).

Besides the Maple Physics ability to handle noncommutative tensor operators and simplify such operators using commutator algebra rules, it is interesting to note: a) the ability of the system to factorize expressions involving products of noncommutative operands (see eqs. **(90)** and **(108)**) and b) the extension of the algorithms for simplifying tensorial expressions [5] to the noncommutativity domain, used throughout this presentation.

It is also worth mentioning how equation labels can reduce the whole computation to entering the main definitions, followed by applying a few commands to equation labels. That approach helps to reduce the chance of typographical errors to a very strict minimum. Likewise, the fact that commands and equations distribute over each other allows cumbersome manipulations to be performed in simple ways, as done, for instance, in eqs. **(8)**, **(9)** and **(13)**.

Finally, it was significantly helpful for us to have the typesetting of results using standard mathematical physics notation, as shown in the presentation above.

Appendix

In this presentation, the input lines are preceded by a prompt **>** and the commands used are of three kinds: some basic Maple manipulation commands, the main Physics package commands to set things and simplify expressions, and two commands of the *Physics:-Library* to perform specialized, convenient, operations in expressions.

The basic Maple commands used

- *interface* is used once at the beginning to set the letter used to represent the imaginary unit (default is I but we used i).
- *isolate* is used in several places to isolate a variable in an expression, for example isolating x in $a\,x + b = 0$ results in $x = -\frac{b}{a}$
- *lhs* and *rhs* respectively get the left-hand side A and right-hand side B of an equation $A = B$
- *subs* substitutes the left-hand side of an equation by the righ-hand side in a given target, for example $subs(A = B, A + C)$ results in $B + C$
- @ is used to compose commands. So $(A@B)(x)$ is the same as $A(B(x))$. This command is useful to express an abstract combo of manipulations, for example as in **(108)** $\equiv$ $(lhs = Factor@rhs)$.

The Physics commands used

- *Setup* is used to set algebra rules as well as the dimension of space, type of metric, and conventions as the kind of letter used to represent indices.
- *Commutator* computes the commutator between two objects using the algebra rules set using *Setup*. If no rules are known to the system, it outputs a representation for the commutator that the system understands.
- *CompactDisplay* is used to avoid redundant display of the functionality of a function.
- *d_*[n] represents the ∂_n tensorial differential operator.
- *Define* is used to define tensors, with or without specifying its components.
- *Dagger* computes the Hermitian transpose of an expression.
- *Normal, Expand, Factor* respectively normalizes, expands and factorizes expressions that involve products of noncommutative operands.
- *Simplify* performs simplification of tensorial expressions involving products of noncommutative factors taking into account Einstein's sum rule for repeated indices, symmetries of the indices of tensorial subexpressions and custom commutator algebra rules.
- *SortProducts* uses the commutation rules set using *Setup* to sort the non-commutative operands of a product in an indicated ordering.

The Physics:-Library commands used

- *Library:-ApplyProductsOfDifferentialOperators* applies the differential operators found in a product to the product operands that appear to its right. For example, applying this command to $p\, V(X)\, m_e$ results in $m_e \cdot p(V(X))$
- *Library:-EqualizeRepeatedIndices* equalizes the repeated indices in the terms of a sum, so for instance applying this command to $L_a^2 + L_b^2$ results in $2 \cdot L_a^2$

References

[1] W. Pauli, "On the hydrogen spectrum from the standpoint of the new quantum mechanics," Z. Phys. **36**, 336–363 (1926)
[2] S. Weinberg, "Lectures on Quantum Mechanics, second edition, Cambridge University Press," 2015.
[3] Veronika Gáliková, Samuel Kováčik, and Peter Prešnajder, "Laplace-Runge-Lenz vector in quantum mechanics in noncommutative space", J. Math. Phys. **54**, 122106 (2013)
[4] Castro, P.G., Kullock, R. "Physics of the $so_q(4)$ hydrogen atom". Theor. Math. Phys. **185**, 1678–1684 (2015).

[5] L. R. U. Manssur, R. Portugal, and B. F. Svaiter, "Group-Theoretic Approach for Symbolic Tensor Manipulation," International Journal of Modern Physics C, Vol. **13**, No. 07, pp. 859-879 (2002).